# $Id$
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Robson Cardoso dos Santos <cardoso.rcs@gmail.com>
# Contributor: Ivan c00kiemon5ter Kanakarakis <ivan.kanak@gmail.com>

pkgname="markdown"
pkgver=1.0.1
pkgrel=5
pkgdesc="Text to (X)HTML conversion tool for web writers."
arch=('any')
url="http://daringfireball.net/projects/markdown/"
license=('BSD')
depends=('perl')
source=("http://daringfireball.net/projects/downloads/Markdown_${pkgver}.zip")
md5sums=('f17b3c2b2830c6fd2fe0098226e59a2f')

build() {
  cd $srcdir/Markdown_$pkgver
  pod2man "Markdown.pl" > "$srcdir/Markdown_$pkgver/$pkgname.1"
}

package() {
  cd $srcdir/Markdown_$pkgver
  install -Dm755 "Markdown.pl" "${pkgdir}/usr/bin/${pkgname}"
  install -Dm644 "License.text" "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
  install -Dm644 "$pkgname.1" "$pkgdir/usr/share/man/man1/$pkgname.1"
}
